

Florian Sikora <florian.sikora@dauphine.fr>

Hello,

here is my solution using MiniZinc <https://www.minizinc.org/>

```
/*minizinc chessarray.mzn --solver gecode -a*/

%games
enum OUTCOME = {left, draw, right};
array[OUTCOME] of var 0..7 : FisherVsKarpov;
array[OUTCOME] of var 0..7 : FisherVsKasparov;
array[OUTCOME] of var 0..7 : KasparovVsKarpov;

%scores
var 0..28: fisher;
var 0..28: kasparov;
var 0..28: karpov;

%7 games
constraint sum(FisherVsKarpov) = 7;
constraint sum(FisherVsKasparov) = 7;
constraint sum(KasparovVsKarpov) = 7;

%points
constraint fisher = FisherVsKarpov[left]*2 + FisherVsKarpov[draw] +
FisherVsKasparov[left]*2 + FisherVsKasparov[draw];
constraint kasparov = FisherVsKasparov[right]*2 + FisherVsKasparov[draw] +
KasparovVsKarpov[left]*2 + KasparovVsKarpov[draw];
constraint karpov = FisherVsKarpov[right]*2 + FisherVsKarpov[draw] +
KasparovVsKarpov[right]*2 + KasparovVsKarpov[draw];

%fisher wins
constraint fisher > kasparov;
constraint fisher > karpov;

%kasparov gets more wins
constraint FisherVsKasparov[right] + KasparovVsKarpov[left] > FisherVsKarpov[left] +
FisherVsKasparov[left]; %than Fisher
constraint FisherVsKasparov[right] + KasparovVsKarpov[left] > FisherVsKarpov[right] +
KasparovVsKarpov[right]; %than karpov

%karpov get less losses
constraint FisherVsKarpov[left] + KasparovVsKarpov[left] < FisherVsKarpov[right] +
FisherVsKasparov[right]; %than fisher
constraint FisherVsKarpov[left] + KasparovVsKarpov[left] < FisherVsKasparov[left] +
KasparovVsKarpov[right]; %then kasparov

solve satisfy;
```

It's very similar to the other solutions, so I'm not sure it's worth to be online.

Best,
Florian Sikora

```
$ minimzinc chessarray.mzn --solver gecode -a FisherVsKarpov = array1d(OUTCOME, [0, 7, 0]);  
FisherVsKasparov = array1d(OUTCOME, [4, 0, 3]); KasparovVsKarpov = array1d(OUTCOME, [2, 4, 1]);  
fisher = 15; kasparov = 14; karpov = 13;
```

```
FisherVsKarpov = array1d(OUTCOME, [0, 7, 0]); FisherVsKasparov = array1d(OUTCOME, [4, 0, 3]);  
KasparovVsKarpov = array1d(OUTCOME, [2, 3, 2]); fisher = 15; kasparov = 13; karpov = 14;
```
